```
> with(OreModules):
> with(OreMorphisms):
> with(Stafford):
> with(linalg):
```

Let us consider the second Weyl algebra $A = A_2(\mathbb{Q})$, where $\mathbb{Q}$ is the field of rational numbers,

```
> A := DefineOreAlgebra(diff=[dx,x], diff=[dy,y], polynom=[x,y],
> comm=[u,rho,c]):
```

and the left A-module L finitely presented by the matrix P of PD defined by:

```
> P := evalm([[u*rho*dx,c^2*dx,0],[0,c^2*dy,u*rho*dx],[rho*dx,u*dx,rho*dy]]);
```

$$P := \begin{bmatrix} u\,\rho\,dx & c^2\,dx & 0 \\ 0 & c^2\,dy & u\,\rho\,dx \\ \rho\,dx & u\,dx & \rho\,dy \end{bmatrix}$$

The left A-module L corresponds to the steady two-dimensional rotational isentropic flow (see R. Courant, D. Hilbert, *Methods of Mathematical Physics*, Volume II, Wiley Classics Library, Wiley, 1962, pp. 436-437). Let us compute $ker_A(.\mathrm{P})$:

```
> SyzygyModule(P,A);
```

$$\mathrm{INJ}(3)$$

Thus, P has full row rank, which shows that $rank_A(A^{1\times 3}\,P) = 3$. Hence, L admits Stafford's reduction. Let us compute one. We obtain

```
> S := map(collect,StaffordReduction(P,A),[dx,dy]):
> nops(S);
```

$$2$$

that L is isomorphic to the left A-module L_1 finitely presented by S[1], where S[1] is defined by

```
> S[1];
```

$$\begin{aligned}
&[0\,,\; u\,\rho\,(-u^2+c^2)\,dx^2 + c^2\,dy^2\,u\,\rho]\\
&[(-u^2+c^2)\,dx^2\,,\; -u\,dx\,\rho\,dy]\\
&[(c^2\,(-u^2+c^2) + c^2\,(-u^2+c^2)\,x)\,dy\,dx - c^2\,(-u^2+c^2)\,dy\,,\\
&-u\,\rho\,dx\,(-u^2+c^2) + (-u\,\rho\,x\,c^2 - c^2\,u\,\rho)\,dy^2]
\end{aligned}$$

and the left A-isomorphism γ from L_1 onto L is induced by S[2], where S[2] is defined by

```
> S[2];
```

$$\begin{bmatrix} u\,\rho\,x + u\,\rho & 1 + c^2\,x + c^2 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

We can check again that γ defines a left A-isomorphism:

```
> TestIso(S[1],P,S[2],A);
```

$$true$$

If $f_1,\ f_2,\ f_3$ are the standard basis vectors of $A^{1\times 3}$, then the residue classes $y_1,\ y_2,\ y_3$ of $f_1,\ f_2,\ f_3$, respectively, form a family of generators of L. Then z = S[2] y is a set of generators { $z_1,\ z_2$ } of L, where y = $(y_1,\ y_2,\ y_3)$^T and z = $(z_1\ z_2)$^T. In particular, we have

$$z_1 = u\rho(x+1)y_1 + (c^2(x+1)+1)y_2, \quad z_2 = y_3.$$

The inverse γ^{-1} of γ is induced by U[1], where U[1] is the first entry of U defined by:

```
> U := map(collect,InverseMorphism(S[1],P,S[2],A),[dx,dy,u,rho,c],distributed);
```

$$U := \left[\left[\begin{array}{cc} \dfrac{dx}{u\,\rho} + \dfrac{(x+1)\,c^2\,dx}{u\,\rho} - \dfrac{c^2}{u\,\rho} & -\dfrac{(1+c^2\,(x+1))\,dy}{-u^2+c^2} \\ 1+(-x-1)\,dx & \dfrac{u\,\rho\,dy\,(x+1)}{-u^2+c^2} \\ 0 & 1 \end{array} \right], \right.$$
$$\left. \left[\begin{array}{ccc} 0 & \dfrac{1}{-u^2+c^2} & 0 \\ 0 & 0 & -\dfrac{1}{-u^2+c^2} \\ 0 & \dfrac{1+c^2\,(x+1)+(-x-1)\,u^2}{u\,(-u^2+c^2)} & 0 \end{array} \right] \right]$$

We can check again that U[1] defines a left A-isomorphism from L to L_1.

```
> TestIso(P,S[1],U[1],A);
```

$$true$$

The above results show that y = U[1] z, and the linear PD system P y = 0 is equivalent to the linear PD system S[1] z = 0 defined by two unknowns and three equations.

One relation can be removed from S[1], i.e., there exists a (2 x 2) matrix T with entries in A such that $A^{1\times 3}$ S[1] = $A^{1\times 2}$ T, and thus L is isomorphic to the left A-module L_1 finitely presented by T. Let us compute such a matrix T:

```
> Q := StaffordReduction(P,A,"reduce_relations"=true):
> nops(Q);
```

$$2$$

We can choose T = Q[1], where:

```
> Q[1];
```

$$\begin{array}{l} [c^2\,(-u^2+c^2)\,dx\,dy\,x - c^2\,(-u^2+c^2)\,dy + c^2\,(-u^2+c^2)\,dx\,dy\,, \\ -\rho\,dy^2\,c^2\,x\,u - u\,\rho\,dx\,(-u^2+c^2) - c^2\,dy^2\,u\,\rho] \\ [(u^2-c^2)\,dx^2\,,\ u\,dx\,\rho\,dy] \end{array}$$

Moreover, we have Q[2] = S[2]

```
> map(collect,Q[2],[u,c,rho]);
```

$$\left[\begin{array}{ccc} u\,\rho\,(x+1) & 1+c^2\,(x+1) & 0 \\ 0 & 0 & 1 \end{array} \right]$$

which shows that the left A-isomorphism γ from L onto L_1 is induced by Q[2].